

Paper 84-28

A Drill-Down Diet: An Example of a Dynamic Detail Access on the Web Using SAS/GRAPH® and ODS

Nadia Redmond, Kaiser Permanente Center for Health Research, Portland, OR

ABSTRACT

With web-based graphs and reports, we have been able to accelerate decision-making for our national multi-site clinical studies. In this paper, I expanded our use of SAS/GRAPH and ODS (Output Delivery System) to create a drill-down profile plot from customized box plots identifying those study participants who are 'outliers'. This paper utilizes the production version of the SAS/GRAPH and ODS features in SAS V8.2.

I assume at least novice knowledge of SAS/GRAPH and ODS.

INTRODUCTION

During a clinical trial study, many monitoring reports and graphs are created to assure the quality of outcome measurements. This paper uses data from Statistics Library at <http://lib.stat.cmu.edu/datasets/bodyfat> that is freely distributed for non-commercial use purposes. With the use of this data, I was able to imitate my previous paper (SUGI26, 2001) that used our national dietary study data.

This data set contains the percentage of body fat, density as determined from underwater weighing, age (years), weight (lbs.), height (inches) and various body circumference measurements for 252 men. With the use of this data, I was able to calculate the body mass index (kg/m²) and create fictitious data for each outlier with 14 weekly measurements to represent the dietary/weight-maintenance study.

The use of box plots in the study determined whether a subject experienced a major weight loss or gain. For the original box plot, the PROC GPLOT used an annotated data set to label the total number of subjects at each clinic site (see Figure 1). The boxes represent 25th percentile, median, and 75th percentile. The whiskers represent 5th and 95th percentiles. The points represent outliers with major weight gain or loss.


After identifying the outliers from the box plot, a profile plot of each subject's weight over the 14-week study period was created for clarification. This paper will demonstrate how to create the box plot graphs that drill-down to new graphs (profile plots) and display more of the details in the data.

To web-enable my plots and graphs, I used ODS HTML output with GIF image format. This format can be easily distributed, as well as be user-friendly. I will demonstrate the static image format to create an HTML document and associated GIF files.

BOX PLOT CODE

The purpose of this paper is to show you how I created the drill-down web enabled box plots using ODS. For detail explanation of the annotate facility for identifying the outliers on the box plot is provided in my previous paper (SUGI26, 2001). All of the key SAS/GRAPH code will be shown, but some of it will not be discussed. The box plot below identifies the outliers in the study and I want the user to be able to point and click on each outlier to drill-down to a profile plot for each subject's weight over the 14-week study period.

Box Plot Example



Boxes represent 25th percentile, median, 75th percentile.
Whiskers represent 1st and 99th percentile, points represent outliers.

C:\SUGI28\SUGI28_PAPER\SAS

Before I can create the ODS and drill-down links on the box plot, I need to create a SAS data set with the linkable variable that will specify the name of another HTML file (the profile plot). This HTML file will be linked and executed when a user presses on the outliers in the box plot.

```
SAS DATA step code:
data dietbmi;
  set sasds.bodyfatbmi;
  if group eq 1 and (id=39 or id=45)
  then
    iddrill='href="Diet1_body.html"';
  else if group eq 2 and (id=41 or
  id=182) then
    iddrill='href="Diet2_body.html"';
run;
```

The variable IDDRILL is my drill-down link in the HTML file. Since I used the Annotate facility to identify the outliers, I must create the HTML variable in my annotate data set to specify the link information for a drill-down graph. This annotate data set is discussed in detail in my previous paper.

```

Annotate Data step code:
%annomac;
proc sort data=dietsbmi out=bmiout;
  by group id weight;
run;
data annobmi;
  length function color style $8;
  retain function 'label' color 'black'
when 'a' style 'swissb' 'xsys' '2' 'ysys'
'2' position '6' size 2.5 'hsys' '3';
  set bmiout;
  by group id bmi;
  x=group;
  y=5;

  if id in (39,41,45,182) then do;
    TEXT=put(ID,3.);
    y=bmi;

/* group identifies the two different
study groups */

    if group=1 then
      HTML=iddrill;
    else if group=2 then
      HTML=iddrill;

output annobmi;
end;
run;

```

SAS ODS CODE

Now, I can begin the ODS code and related options necessary to web enable the box plot and the drill-down link. Let's first look at the code below.

```

filename odsout 'c:\sugi28\HTML Output';
goptions reset=global xpixels=650
          ypixels=450 device=gif;
ods html body='boxplot_body.html'
  (title="ODS Drill-Down Boxplot")
  frame='boxplot_frame.html'
  contents='boxplot_contents.html'
  path = odsout
  gtitle gfootnote
  gpath='c:\sugi28\HTML Output'
  parameters=("ddlevel1"
"disabledrilldown"="true"
"drilldownmode"="any");

/* SAS/GRAPH code here */

ods html close;

```

The above code begins with the familiar ODS HTML statement that directs the output to a body file called 'boxplot_body.html.' All my ODS output files will be stored at 'c:\sugi28\HTML Output' folder. The ODS statement above contains a few important options. I have selected GTITLE and GFOOTNOTE to tell SAS/GRAPH that I have defined my titles and footnotes and to place them in the resulting GIF file. The GPATH option is used with DEVICE=HTML, GIF or JPEG. It tells SAS/GRAPH where to store the GIF file and I want GIF file to be stored in the same location as the HTML file. The parameters= option controls the drill-down capabilities. I have used the options that Curtis Smith (SUGI27, 2002) demonstrated in

his paper. He describes these options and others in detail. The options that I specified are:

DdLeveln

Specifies internal drill-down level specifications.

DisableDrillDown

Disables the internal JAVA drill-down when using SAS/GRAPH procedure options HTML and HTML_Legend to create drill-down graphs.

DrillDownMode

Specifies the drill-mode to use when drilling down. The values are "Any", "Local", "Script", "URL" and "HTML". I used the default "Any". In this mode, the graph first attempts to drill-down using "Local" mode. If it fails, then JavaScript ("Script") is used. If both fail, "URL" mode is attempted. If all fail, "HTML" mode is used.

GOPTIONS STATEMENT AND DEVICE

The GOPTIONS statement with the DEVICE= option is important to determine the type of output you want. I chose DEVICE=GIF that SAS will create an HTML file with a link to a GIF file that is a static image of the graph.

The XPIXEL=650 and YPIXEL=450 worked the best. The larger the number of pixels the bigger the graphics image will be.

SAS/GRAPH HTML OPTION

It is very important that the HTML option of the SAS/GRAPH procedure is used. I used this option to make the drill-down capability work. The HTML= option identifies the variable in the input data set whose values create links in the HTML file (see above DATA step and Annotate data set). The links are associated with an area on the plot and point to the data or graph that is displayed (drilled down) when the user points and clicks on the area.

Let's take a look at the Box Plot code I created with ANNOTATE option in PROC GPLOT.

```

goptions reset=(axis, legend, pattern,
symbol, title, footnote)
          rotate=landscape hpos=0 vpos=0
          htext= ftext= ctext= target=
          gaccess= gsfmtode= ;
goptions ctext=BLACK ftext=SWISSB
          graphrc interpol=join NOPROMPT;
symbol1 c=BLUE i=BOX01 v=dot h=1
        cv=RED;
axis1 color=BLACK width=1.0 style=1
      major=none minor=none ORDER=(1 TO
2 BY 1) OFFSET=(4 CM);
axis2 color=BLACK width=2.0 style=1
      LABEL=NONE;

proc gplot data=dietsbmi;
  plot bmi * group / haxis=axis1
    vaxis=axis2 annotate=annobmi;
  format group grpfmt.;
  TITLE h=2 "BMI (kg/m**2) - Males";
  FOOTNOTE H=1.2 J=L "Boxes represent
    25th percentile, median, 75th
    percentile.";
  FOOTNOTE2 H=1 J=L "Whiskers

```

```


represent 1st and 99th
percentile, points represent
outliers.";
FOOTNOTE4 H=.7 J=L "&program" j=r
"&SYSDATE";
run; quit;

```

I used many SAS/GRAPH options to create the box plot as shown above (page 1). These options are explained in detail in my previous paper (SUGI26, 2001).

Moving the mouse over the outlier points on the box plots changes the mouse shape to a hand, depending on the Windows scheme you are using. Once you get a hand, then you can press the mouse button to drill down into another related graph. Below is one of the profile plots displaying Diet I group outliers.

The Profile Plot for Diet I Example



Profile Plots of Individual Outliers – for Diet I

C:\SUGI28\SUGI28_PAPER.SAS

This profile plot was created with following SAS/GRAPH code:

```

goptions reset=(axis, legend, pattern,
symbol, title, footnote)
rotate=landscape hpos=0 vpos=0
htext= ftext= ctext= target=
gaccess= gsfmode= ;
goptions ctext=BLACK ftext=SWISSB
graphrc interpol=join
NOPROMPT;
symbol1 c=DEFAULT i=join v=+ h=1.2
cv=RED l=1;
symbol2 c=DEFAULT i=join v=- h=1.2
cv=BLUE l=21;
symbol3 c=DEFAULT i=join v=: h=1.2
cv=GREEN l=1;
symbol4 c=DEFAULT i=join v=$ h=1.2
cv=BLUE l=21;
axis1 color=BLACK width=2.0 style=1
major=none minor=none order=(175
to 375 by 20) label=(a=90 h=1.5
'Weight in lbs');
axis2 color=BLACK width=2.0 style=1
offset=(2) LABEL=(h=1.5 'Week');

```

```

%let diet1=Diet I;
%let diet2=Diet II;

%macro gplot_m;
%do I=1 %to 2;
ods html body="Diet&I._body.html"
path=odsout;
proc gplot data=sasds.outlier;
where group=&I;
plot weight * week=ID /
haxis=axis2 vaxis=axis1
skipmiss;
TITLE h=2 "Weight (lbs) - Outliers";
FOOTNOTE H=1.5 J=L "Profile Plots of
Individual Outliers - for
&&diet&I";
FOOTNOTE2 H=1 J=L "&program" j=r
"&SYSDATE";

run;
%end;
quit;
%mend gplot_m;
%gplot_m;
ods html close;
ods listing;

```

CONCLUSION

It is not possible to show or explain all the features of SAS/GRAPH and ODS in this paper. I have found that SAS for Windows on-line help for SAS/GRAPH and ODS is very complete.

SAS/GRAPH software with ODS produces presentation quality custom graphic, especially for the web. The production of graphics to address specific health-measurement quality checks can be done very efficiently, especially where many sets of these graphics are needed regularly. The BOXPLOT procedure (new in SAS version 8) could have been used in this paper, but it did not provide me the analytical capabilities of Gplot to be more specific with the whiskers and percentiles.

REFERENCES

Redmond, Nadia. (2001), "Data Visualization of Outliers from a Health Research Perspective Using SAS/GRAPH and the Annotate Facility", in *Proceedings of the Twenty-sixth Annual SUGI Conference*, Cary, N.C.: SAS Institute Inc. pp. 70-74.

Smith, Curtis A.. (2002), "Creating Drill-Down Graphs Using SAS/GRAPH and the Output Delivery System", in *Proceedings of the WUSS Annual Conference*, Cary, N.C.: SAS Institute Inc.

ACKNOWLEDGMENTS

SAS and all other SAS Institute Inc. product or service names are registered trademarks or trademarks of SAS Institute Inc. in the USA and other countries. ® indicates USA registration.

Other brand and product names are trademarks of their respective companies.

CONTACT INFORMATION

Your comments and questions are valued and encouraged. Contact the author at:

Nadia Redmond

Kaiser Permanente Center for Health Research

3800 N. Interstate Avenue

Portland, OR 97227

Email: Nadia.Redmond@kpchr.org

Web: www.kpchr.org